

## Geometría Diferencial MATE2410

## Laboratorios de Maple

## Lab4: Geodésicas

## 1. Objetivos:

Usar Maple para hacer muchos cálculos algebraicos necesarios para hallar geodésicas.

## 2. Geodésicas

## 2.1. Cálculos básicos

```

> restart: with(plots): with(LinearAlgebra):
> EFG:=proc(X)# PRIMERA FORMA FUNDAMENTAL
> local Xu,Xv,E,F,G;
> Xu:=<diff(X[1],u),diff(X[2],u),diff(X[3],u)>;
> Xv:=<diff(X[1],v),diff(X[2],v),diff(X[3],v)>;
> E:=DotProduct(Xu,Xu,conjugate=false);
> F:=DotProduct(Xu,Xv,conjugate=false);
> G:=DotProduct(Xv,Xv,conjugate=false);
> simplify([E,F,G]);
> end:
> UN:=proc(X) # VECTOR NORMAL
> local Xu,Xv,Z,s;
> Xu:=<diff(X[1],u),diff(X[2],u),diff(X[3],u)>;
> Xv:=<diff(X[1],v),diff(X[2],v),diff(X[3],v)>;
> Z:=CrossProduct(Xu,Xv);
> s:=VectorNorm(Z,Euclidean,conjugate=false);
> simplify(<Z[1]/s|Z[2]/s|Z[3]/s>,sqrt,trig,symbolic);
> end:
> lmn:=proc(X) # SEGUNDA FORMA FUNDAMENTAL
> local Xu,Xv,Xuu,Xuv,Xvv,U,l,m,n;
> Xu:=<diff(X[1],u),diff(X[2],u),diff(X[3],u)>;
> Xv:=<diff(X[1],v),diff(X[2],v),diff(X[3],v)>;
> Xuu:=<diff(Xu[1],u),diff(Xu[2],u),diff(Xu[3],u)>;
> Xuv:=<diff(Xu[1],v),diff(Xu[2],v),diff(Xu[3],v)>;
> Xvv:=<diff(Xv[1],v),diff(Xv[2],v),diff(Xv[3],v)>;
> U:=UN(X);
> l:=DotProduct(U,Xuu,conjugate=false);
> m:=DotProduct(U,Xuv,conjugate=false);
> n:=DotProduct(U,Xvv,conjugate=false);
> simplify([l,m,n],sqrt,trig,symbolic);
> end:
> MK:=proc(X)# CURVATURA MEDIA
> local E,F,G,l,m,n,S,T;
> S:=EFG(X); T:=lmn(X);
> E:=S[1]; F:=S[2]; G:=S[3];
> l:=T[1]; m:=T[2]; n:=T[3];
> simplify((G*l+E*n-2*F*m)/(2*E*G-2*F^2),sqrt,trig,symbolic);
> end:
> GK:=proc(X)# CURVATURA DE GAUSS
> local E,F,G,l,m,n,S,T;
> S:=EFG(X); T:=lmn(X);
> E:=S[1]; F:=S[2]; G:=S[3];
> l:=T[1]; m:=T[2]; n:=T[3];
> simplify((1*n-m^2)/(E*G-F^2),sqrt,trig,symbolic);
> end:

```

```

> shape:=proc(X)# OPERADOR FORMA
> local Y,Z,a,b,c,d;
> Y:=EFG(X);
> Z:=lmn(X);
> a:=simplify((Z[1]*Y[3]-Z[2]*Y[2])/(Y[1]*Y[3]-Y[2]^2));
> b:=simplify((Z[2]*Y[1]-Z[1]*Y[2])/(Y[1]*Y[3]-Y[2]^2));
> c:=simplify((Z[2]*Y[3]-Z[3]*Y[2])/(Y[1]*Y[3]-Y[2]^2));
> d:=simplify((Z[3]*Y[1]-Z[2]*Y[2])/(Y[1]*Y[3]-Y[2]^2));
> [S(x_u)=a*x_u+b*x_v,S(x_v)=c*x_u+d*x_v];
> end:
> shape_matrix:=proc(X)# MATRIZ OPERADOR FORMA
> local Y,Z,a,b,c,d;
> Y:=EFG(X);
> Z:=lmn(X);
> a:=simplify((Z[1]*Y[3]-Z[2]*Y[2])/(Y[1]*Y[3]-Y[2]^2));
> b:=simplify((Z[2]*Y[1]-Z[1]*Y[2])/(Y[1]*Y[3]-Y[2]^2));
> c:=simplify((Z[2]*Y[3]-Z[3]*Y[2])/(Y[1]*Y[3]-Y[2]^2));
> d:=simplify((Z[3]*Y[1]-Z[2]*Y[2])/(Y[1]*Y[3]-Y[2]^2));
> Matrix([[a,c],[c,d]]);
> end:

```

Warning, the name changecoords has been redefined

## 2.2. Ecuaciones de las geodésicas

```

> geoeq:=proc(X)
> local M,eq1,eq2;
> M:=EFG(X);
> eq1:=diff(u(t),t$2)+subs({u=u(t),v=v(t)
> },diff(M[1],u)/(2*M[1]))*diff(u(t),t)^2
> +subs({u=u(t),v=v(t)},diff(M[1],v)/(M[1]))*diff(u(t),t)*diff(v(t),t)
> -subs({u=u(t),v=v(t)},diff(M[3],u)/(2*M[1]))*diff(v(t),t)^2=0;
> eq2:=diff(v(t),t$2)-subs({u=u(t),v=v(t)
> },diff(M[1],v)/(2*M[3]))*diff(u(t),t)^2
> +subs({u=u(t),v=v(t)},diff(M[3],u)/(M[3]))*diff(u(t),t)*diff(v(t),t)
> +subs({u=u(t),v=v(t)},diff(M[3],v)/(2*M[3]))*diff(v(t),t)^2=0;
> eq1,eq2;
> end:

```

## 2.3. Ejemplos

```

> sphere:=<cos(u)*cos(v)|sin(u)*cos(v)|sin(v)>;
> EFG(sphere);
> geoeq(sphere);

```

$$\begin{aligned}
 sphere &:= [\cos(u) \cos(v), \sin(u) \cos(v), \sin(v)] \\
 & \quad [\cos(v)^2, 0, 1]
 \end{aligned}$$

$$\left(\frac{\partial^2}{\partial t^2} u(t)\right) - \frac{2 \sin(v(t)) \left(\frac{\partial}{\partial t} u(t)\right) \left(\frac{\partial}{\partial t} v(t)\right)}{\cos(v(t))} = 0, \quad \left(\frac{\partial^2}{\partial t^2} v(t)\right) + \cos(v(t)) \sin(v(t)) \left(\frac{\partial}{\partial t} u(t)\right)^2 = 0$$

```

> torus:=<(4+cos(u))*cos(v)|(4+cos(u))*sin(v)|sin(u)>;
> EFG(torus);
> geoeq(torus);

```

$$\begin{aligned}
 torus &:= [(4 + \cos(u)) \cos(v), (4 + \cos(u)) \sin(v), \sin(u)] \\
 & \quad [1, 0, 16 + 8 \cos(u) + \cos(u)^2]
 \end{aligned}$$

$$\begin{aligned}
 \left(\frac{\partial^2}{\partial t^2} u(t)\right) - (-4 \sin(u(t)) - \cos(u(t)) \sin(u(t))) \left(\frac{\partial}{\partial t} v(t)\right)^2 &= 0, \\
 \left(\frac{\partial^2}{\partial t^2} v(t)\right) + \frac{(-8 \sin(u(t)) - 2 \cos(u(t)) \sin(u(t))) \left(\frac{\partial}{\partial t} u(t)\right) \left(\frac{\partial}{\partial t} v(t)\right)}{16 + 8 \cos(u(t)) + \cos(u(t))^2} &= 0
 \end{aligned}$$

```

> witch:=<2*tan(u)|2*cos(u)^2*cos(v)|2*cos(u)^2*sin(v)>;
> EFG(witch);
> geoeq(witch);

```

$$\text{witch} := [2 \tan(u), 2 \cos(u)^2 \cos(v), 2 \cos(u)^2 \sin(v)]$$

$$[-4 \frac{-1 - 4 \cos(u)^6 + 4 \cos(u)^8}{\cos(u)^4}, 0, 4 \cos(u)^4]$$

$$\left( \frac{\partial^2}{\partial t^2} u(t) \right) - \frac{1}{8} \frac{\left( -4 \frac{24 \cos(u(t))^5 \sin(u(t)) - 32 \cos(u(t))^7 \sin(u(t))}{\cos(u(t))^4} - \frac{16 \%1 \sin(u(t))}{\cos(u(t))^5} \right) \cos(u(t))^4 \left( \frac{\partial}{\partial t} u(t) \right)^2}{\%1}$$

$$- \frac{2 \cos(u(t))^7 \sin(u(t)) \left( \frac{\partial}{\partial t} v(t) \right)^2}{\%1} = 0, \left( \frac{\partial^2}{\partial t^2} v(t) \right) - \frac{4 \sin(u(t)) \left( \frac{\partial}{\partial t} u(t) \right) \left( \frac{\partial}{\partial t} v(t) \right)}{\cos(u(t))} = 0$$

$$\%1 := -1 - 4 \cos(u(t))^6 + 4 \cos(u(t))^8$$

## 2.4. Gráficos de las geodésicas

```

> plotgeo:=proc(X,ustart,uend,vstart,vend,u0,v0,Du0,Dv0,T,N,gr,theta,phi)
> i)
> local sys,desys,u1,v1,listp,geo,plotX;
> sys:=geoeq(X);
> desys:=dsolve({sys,u(0)=u0,v(0)=v0,D(u)(0)=Du0,D(v)(0)=Dv0},
> {u(t),v(t)},type=numeric,output=listprocedure);
> u1:=subs(desys,u(t)); v1:=subs(desys,v(t));
> geo:=tubeplot(convert(subs(u='u1'(t),v='v1'(t),X),list),
> t=0..T,radius=0.05,color=black,numpoints=N):
> plotX:=plot3d(subs({u=u(t),v=v(t)
> },X),u=ustart..uend,v=vstart..vend,grid=[gr[1],
> gr[2]],shading=XY,lightmodel=light3):
> display({geo,plotX
> },style=patch,scaling=constrained,orientation=[theta,phi]);
> end:


```

### 2.4.1. Ejemplos

```

> plotgeo(torus,0,2*Pi,0,2*Pi,Pi/2,0,0,1,30,150,[15,30],105,33);


```




```

> plotgeo(torus,0,2*Pi,0,2*Pi,0,0,0,1,15,75,[15,30],177,68);


```




```
> plotgeo(torus,0,2*Pi,0,2*Pi,0,0,1,15,165,[15,30],177,68);
```




```
> plotgeo(torus,0,2*Pi,0,2*Pi,0,0,1,35,165,[15,30],177,68);
```



```
> plotgeo(sphere,0,2*Pi,0,2*Pi,0,0,1,1,5,35,[15,30],177,68);
```



```
> plotgeo(witch,-1.1,1.1,0,2*Pi,Pi/3,0,0,1,100,200,[15,30],92,-113);
```



## 2.5. Ejercicios

1. Considere el hiperboloide de un solo manto, como superficie de revolución sobre el eje  $z$ . Halle la métrica, las ecuaciones diferenciales que determinan las geodésicas y grafique diferentes geodésicas, entre otras grafique una geodésica que sea un meridiano y una geodésica que compruebe que un paralelo en particular no es geodésica. Ayuda: Vea la Figura 5.19 de la página 206, la cual está hecha para un cono.

2. Considere un cilindro vertical. En este caso los paralelos y los meridianos si son geodésicas. Compruébelo usando Maple. Además haga dibujos de geodésicas que no sean ni paralelos ni meridianos. Cómo son estas curvas?. Compruébelo resolviendo las ecuaciones.

## 3. Manejo del comando plotgeo

El objetivo es dibujar una superficie  $M$  con una parametrización ortogonal  $X$ , es decir tal que  $F = 0$ , y sobre ella una geodésica  $\alpha$ .

Para resolver el sistema de ecuaciones diferenciales ordinarias de segundo orden debemos dar las condiciones iniciales: un punto  $p \in M$  y un vector  $\mathbf{v} \in T_p M$ . El punto  $p = X(u_0, v_0)$ , la curva geodésica  $\alpha(t) = X(u(t), v(t))$  y el vector  $\mathbf{v} = \alpha'(t) = X_u u' + X_v v'$  el cual da la dirección inicial se establece dando

```
plotgeo(X,ustart,uend,vstart,vend,u0,v0,Du0,Dv0,T,N,gr,theta,phi);
plotgeo(witch,-1.1,1.1,0,2*Pi,Pi/3,0,0,1,100,200,[15,30],92,-113);
Este comando tiene 14 entradas a saber:
```

1. Nombre de la superficie a graficar:  $X = \text{witch}$

2.  $u_{min} \leq u \leq u_{max}$   $u_{min} = \mathbf{ustart} = -1,1$
3.  $u_{max} = \mathbf{uend} = 1,1$
4.  $v_{min} \leq v \leq v_{max}$   $v_{min} = \mathbf{vstart} = 0$
5.  $v_{max} = \mathbf{vend} = 2*\mathbf{Pi}$
6. Primera coordenada del punto  $p$ ,  $u_0 = u0 = \mathbf{Pi}/3$
7. Segunda coordenada del punto  $p$ ,  $v_0 = v0 = 0$
8. Primera componente del vector  $\mathbf{v}$ ,  $\mathbf{Du0}=0$
9. Segunda componente del vector  $\mathbf{v}$ ,  $\mathbf{Dv0}=1$
10. El valor de  $T$  es el valor numérico máximo del parámetro  $t$  de la geodésica  $\alpha(t)$ ,  $0 \leq t \leq T$ .  
 $\mathbf{T}=100$
11. El valor de  $N$  es el número de puntos cuando se plotea la geodésica, hace que se vea más suave cuando este valor es grande, pero exige máquina, pues debe hacer muchos cálculos.  
 $\mathbf{N}=200$
12. Lo que está dentro del corchete es la grilla en el dominio  $D$  de  $(u, v)$ . Entre más grande mejor pero exige máquina para hacer tanto cálculo.  $\mathbf{gr} = [15, 30]$
13. Para visualizar la gráfica se da un punto de vista (donde está nuestro ojo), en coordenadas esféricas  $\theta = 92$
14. Es el valor de  $\phi$  en coordenadas esféricas donde está nuestro ojo.  $\phi = -113$

Esos últimos valores no son tan importantes aunque siempre se deben colocar. Recuerde que usted puede mover y rotar el gráfico. Cuando hace una traducción a otro formato por ejemplo `.pdf` aparecerá tal cual aparece en primera instancia.

### 3.1. Ecuaciones

1. Hiperboloide de un solo manto

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} - \frac{z^2}{c^2} = 1 \quad X(u, v) = (a \cosh u \cos v, b \cosh u \sin v, c \sinh u)$$

2. Cilindro vertical

$$\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1 \quad X(u, v) = (a \cos v, b \sin v, u)$$